

J Quick Tutorial – 1, 数値計算とプログラミング J-Quick Reference を使って 西川 利男

今回と次回にわたって、あらためて初心に戻って、まずは数値計算とそのプログラミングから、Jの基本を始めてみようと思う。

1. J-Quick Reference をどう使うか

Jのプログラムは分かりにくいという。J-Quick Referenceがそのためにあるが、必ずしも分かりやすく書かれてはいない。

J-Quick Referenceはページを追って、読むためのものではなく、手元において、必要なときに参照するためのものである。

2. Jをまず関数電卓のように動かしてみる

```
√ 2          %: 2 => 1.4142
```

```
π/6          1r6p1=> 0.523599
```

```
sin π/6     1 o. 1r6p1 => 0.5
```

次のようにすれば、三角関数も使いやすくなる。

```
require 'trig'  
sin π/6     sin pi%6=>0.5
```

3. さらに、Jは次のように対話的に使われる。

```
X =: *: 3  
Y =: *: 4  
X  
9  
Y  
16  
X + Y  
25  
  
%: X + Y  
5
```

4. Jは自然言語と同じに扱う

名詞(0) … 値、数値、文字、ボックス

動詞(3) … 演算、関数、操作、処理

それ以外に副詞(1)、接続詞(2)がある。かつこ内の数字は品詞の種別番号を示す。そして、Jシステムにはじめから備えられているものはプリミティブと呼ばれる。その使用例などはJ-Quick Referenceで説明されている。また、ユーザが名前を付けて定義して使うことができる。これがJのプログラミングになる。

なお、便利のためにつぎのような英語名のコマンド、キーワードが備えられている。

names, load, require, clear, ...

5. Jの数値計算プログラム—Explicit(明示定義)と定義関数の実行

Jの動詞の定義、つまりプログラムには、いろいろな方式があるが、まずは次のExplicit(明示定義)で行う。

例1 標準正規分布関数の関数値

$$f(x) = \frac{1}{\sqrt{2\pi}} \cdot \exp\left(-\frac{x^2}{2}\right)$$

5. 1 動詞を示す3で、関数名 normf として定義する。

引数 y (x ではない!) に対して、Jの実行の順序 (右から) の通りプリミティブで記述する。

```
normf =: 3 : 0
```

```
(% %: o. 2) * ^ - -: *: y
```

```
NB. (逆数、平方根、2 π) * exp - 1/2 二乗 引数  
)
```

NB. は注釈、実行されない。最後は右かつこで終わる。

つぎのようにして、プログラム(=定義)は実行されて、関数値が得られる。

```
normf 0  
0.398942  
normf 0.5  
0.352065  
normf 1  
0.241971
```

ところで、複数の引数に対してそれぞれ別々に関数値を求めるには、どうすればよいか。

5. 2 ループ制御構文 — BASIC などでも行われる方法

```
wr =: 1! : 2&2 NB. write value on the screen
normflp =: 3 : 0
i =. 0
while. i < 3
  do.
    yi =. i{y.
    wr yi, normf yi
    i =. i + 1
  end.
)
normflp 0 0.5 1
0 0.398942
0.5 0.352065
1 0.241971
3
```

5. 3 動詞のランクを指定して実行 — J独自の非常に強力な方法

動詞のランク 0 を指定すると、複数の引数をそれぞれ単独な値として実行する。

```
normf" (0) 0 0.5 1
0.398942 0.352065 0.241971
```

例 2 標準正規分布関数の積分値

$$\begin{aligned}\phi(x) &= \frac{1}{\sqrt{2\pi}} \int_0^x \exp\left(-\frac{t^2}{2}\right) dt \\ &= \frac{x}{\sqrt{2\pi}} \left[1 - \frac{x^2}{3 \cdot 2 \cdot 1!} + \frac{x^4}{5 \cdot 2^2 \cdot 2!} - \frac{x^6}{7 \cdot 2^3 \cdot 3!} + \dots \right]\end{aligned}$$

岩波数学公式 I p.155 より

定義は以下のように一行で書く必要はない。分かりやすく書くことが第一である。

```
intnormf =: 3 : 0
10 intnormf y
:
n =. i. x
A =. 2*n
B =. 1 + 2*n
C =. 2^ n
```

```

D =. ! n
S =. (_1 ^ n) * (y ^ A) % (B * C * D)
y * (% %: o.2) * +/S
)

```

実行すると、つぎのようになる。

```
intnormf 0.5
```

0.191462

```
intnormf 1
```

0.341345

```
intnormf 1.5
```

0.433193

この動詞定義では、右引数として x を、左引数として y をとるように定義してある。左引数には級数近似の項数を指定できるようにしてある。

```
4 intnormf 2      近似の項数が不十分
```

0.433137

```
10 intnormf 2      近似の項数が十分 数表の数値と一致
```

0.477241

```
intnormf 2      左引数を省略したときは、10としたものとされる。
```

0.477241


前と同様に、動詞のランクを指定して実行すれば、次のようになる。

```
intnormf"(0) 0.5 1 1.5 2
```

0.191462 0.341345 0.433193 0.477241

6. 簡易プロットツールplotによる標準正規分布関数のグラフ表示

```
require 'plot'  
i: 5  
_5 _4 _3 _2 _1 0 1 2 3 4 5  
normf i: 5  
1.48672e_6 0.00013383 0.00443185 0.053991 0.241971 0.398942 0.241971  
0.053991 0.00443185 0.00013383 1.48672e_6  
plot normf i: 5
```



```
plot normf (i: 50) % 50
```

